

1 AMPL – komputerowy język opisu zadań PL

1.1 Uniwersalny język modelowania i algorytm rozwiązywania

Dwa klucze do sukcesu w badaniach operacyjnych

- język pozwalający opisać szeroką klasę zadań optymalizacyjnych oraz
- algorytm pozwalający rozwiązać wszystkie zadania opisane w tym języku.

Przykładem jest programowanie liniowe

- język wyrażeń liniowych, funkcji, równań i nierówności,
- algorytm simplex (simplex) wraz dodatkowymi metodami.

Komputerowe języki modelowania zadań LP

- American Mathematical Programming Language (**AMPL**) – najpopularniejszy.
- na laboratorium korzystamy z **GNU Linear Programming Kit (GLPK)**, zawierającego język **GNU MathProg**, który jest podzbiorem AMPL,
- są i inne języki, m.in. GAMS, MPL, Lingo.

Inne przykłady języków i algorytmów

- programowanie dynamiczne: duża klasa zadań rozwiązywanych podobnym algorytmem, ale nie ma komputerowego języka opisywania takich zadań.
- programowanie ograniczeń: wyrażenia liniowe i logiczne zapewniają ogromne możliwości definiowania zadań, ale algorytmy są jeszcze mało efektywne.

1.2 Ogólna postać zadania PL w języku AMPL

```
1 # Deklaracje zbiorów
2 ...
3 # Deklaracje parametrów
4 ...
5 # Deklaracje zmiennych
6 ...
7 # Polecenie funkcji celu
8 ...
9 # Polecenia ograniczeń
10 ...
11 # Sekcja danych
12 data;
13 ...
14 end;
```

1.3 Model AMPL dla zadania z przykładu 1.

Deklaracje zmiennych

```
1 # Komentarze po hash 'ach
2 set Wyroby; # Deklaracje zbiorów
3 set Stadia;
4
5 # Deklaracje parametrów
6 param zysk {Wyroby} >= 0;
7 param czas_pracy {Stadia} >= 0;
8 param obciazenie {Stadia, Wyroby} >= 0;
9
10 # Deklaracje zmiennej
11 var liczba {Wyroby} >= 0;
```

Funkcja celu i ograniczenia

```
1 maximize calkowity_zysk:
2     sum{wyrob in Wyroby} zysk[wyrob]*liczba[wyrob];
```

Maksymalizuj całkowity zysk:

sumę po wszystkich wyrobach ze zbioru Wyroby
iloczynu zysku i liczby danego wyrobu.

```
3 subject to max_obciazenie {stadium in Stadia}:
4     sum {wyrob in Wyroby}
5         obciazenie[stadium, wyrob]*liczba[wyrob]
6     <= czas_pracy[stadium];
```

Dla każdego stadium ze zbioru Stadia:

suma po wszystkich wyrobach ze zbioru Wyroby
iloczynu obciążenia i liczby danego wyrobu
musi być mniejsza lub równa czasowi pracy stadium.

Dane

```
1 data;
2     set Wyroby := zszywacz dziurkacz;
3     set Stadia := produkcja malowanie montaz;
4     param zysk :=
5         zszywacz      12
6         dziurkacz     8;
7     param czas_pracy :=
8         produkcja     80
9         malowanie     100
10        montaz        75;
11    param obciazenie:
12        zszywacz      dziurkacz      :=
13        produkcja     4                2
14        malowanie     2                3
15        montaz        5                1;
16 end;
```

1.4 Podstawy AMPL

American Mathematical Programming Language

Na zajęciach używamy solvera, t.j. programu rozwiązującego zadania, **GNU Linear Programming Kit** z interpreterem języka **MathProg** będącego podzbiorem **AMPL**.

Literatura:

1. GNU Linear Programming Kit, *Reference Manual*, Version 4.0, Draft Edition, 2003. (<http://gnuwin32.sourceforge.net/downlinks/glpk-doc.php>)
2. R. Fourer, D.M.Gay, B.W. Kemighan, *AMPL - A Modeling Language for Mathematical Programming*, The Scientific Press, Danvers MA, 1993.

1.4.1 Zbiory nieuporządkowane

```
1 # Deklaracja zbioru nieuporządkowanego
2 # i jego elementów w modelu
3 set DniTygodnia := {"Po","Wt","Sr","Cz","Pi","So","Ni"};
4 # Deklaracja zbioru z elementami w sekcji danych,
5 # jako podzbiór innego zbioru
6 set DniWolne within DniTygodnia;
7 # Zbiór wyliczany jako różnica innych zbiorów
8 set DniRobocze := DniTygodnia diff DniWolne;
9 # Drukowanie wyliczonych zbiorów i parametrów
10 display DniRobocze;
11 data;
12     set DniWolne := So Ni;
13 end;
```

```
1 Start of display output
2 Display statement at line 9
3 DniRobocze: Po Wt Sr Cz Pi
4 End of display output
```

1.4.2 Zbiory uporządkowane i rzadkie wyliczane

```
1 # Parametr skalarny, całkowitoliczbowy, dodatni
2 param LiczbaOperacji integer >0;
3 # Deklaracja wyliczanego zbioru uporządkowanego
4 # tj. zbioru liczb całkowitych
5 set Operacje := 1..LiczbaOperacji by 2;
6 # Deklaracja zbioru złożonego (pochodnego)
7 # jako wynik wyrażenia ‘indeksowego’
8 set Pary := {j in Operacje, k in Operacje: j < k};
9 set Pary := {(j,k) in Operacje cross Operacje: j < k};
10 # Wyrażenie  $j < k$  możliwe tylko dla zbioru uporządkowanego
11
12 display Operacje, Pary;
13 data; param LiczbaOperacji := 8; end;
```

```
1 Start of display output
2 Operacje:  1   3   5   7
3 Pary: (1,3) (1,5) (1,7)
4         (3,5) (3,7)
5         (5,7)
```

1.4.3 Zbiory rzadkie enumerowane

```
1 param LiczbaWezlow integer >0;
2 set Wezly := 1..LiczbaWezlow by 2;
3
4 # Deklaracja zbioru złożonego – rzadkiego
5 set Luki1 within Wezly cross Wezly;
6 set Luki2 within {i in Wezly, j in Wezly:  $i \triangleleft j$ };
7 set Luki3 within {Wezly, Wezly};
8
9 data;
10   param LiczbaWezlow := 8;
11   # Enumeracja (wypisanie) wszystkich elementów
12   set Luki1 := (1,3) (1,5) (3,5);
13   set Luki2 := 1 3 1 5 3 5;
14   set Luki3 : 1 3 5 7 :=
15               1  -  +  +  -
16               3  -  -  +  -
17               5  -  -  -  -
18               7  -  -  -  -  ;
19 end;
```

1.4.4 Parametry i zmienne

```
1 set N := {"op1", "op2", "op3", "op4"}; # Operacje
2 set Luki within N cross N; # Kolejność wykonywania
3
4 # Parametr — wektor zdefiniowany na zbiorze N, dodatni
5 param p {N} > 0; # Czas wykonywania
6 param a {N} >= 0; # Liczba pracowników
7
8 # Parametr o wartościach nie mniejszych od poprzedniego
9 param d {j in N} >= p[j]; # Termin zakończenia
10 param s {Luki} >= 0; # Czas przebrojenia
11
12 # Zmienna — wektor, nieujemna
13 var c {N} >= 0;
14
15 display Luki, a;
16
17 data;
18 # Jeżeli nie podano inaczej to wartość 0
19 param default 0: a := op2 3 op4 4;
20 # Tabełacyjne zestawienie danych
21 # zdefiniowanych na tym samym zbiorze
22 param: p d :=
23 op1 4 5
24 op2 2 7
25 op3 3 6
26 op4 1 4;
27 # Dane dla zbioru Luki oraz parametru s
28 param: Luki: s :=
29 op1 op2 3
30 op1 op4 1
31 op2 op3 2;
32 end;
```

```
1 Luki: (op1, op2) (op1, op4) (op2, op3)
2 a[op2] = 3 a[op4] = 4
```

1.4.5 Funkcja celu i ograniczenia

$$\min \sum_{j \in N} c_j x_j \quad \text{--- Minimalizacja danej funkcji celu}$$

```
1 minimize FunkcjaCelu:
2     sum {j in N} c[j] * x[j];
```

$$\sum_{i \in M} \sum_{j \in N} a_{ij} y_{ij} \geq d \quad \text{--- Jedno ograniczenie}$$

```
3 subject to DotrzymanieJednegoWarunku:
4     sum {i in M, j in N}
5         a[i, j] * y[i, j] >= d;
```

$$\sum_{j \in N} a_{ij} y_{ij} \geq b_i, \quad \forall i \in M \quad \text{--- Grupa podobnych ograniczeń}$$

```
6 s.t. DotrzymanieWarunkuDlaKazdego {i in M}:
7     sum {j in N}
8         a[i, j] * y[i, j] >= b[i];
```

$$y_{ij} \geq b_i, \quad \forall i \in M, \forall j \in N \quad \text{--- Grupa ograniczeń}$$

```
9 s.t. DotrzymanieWarunkuDlaKazdejPary {i in M, j in N}:
10    y[i, j] >= b[i];
```

Zamiast **subject to** wystarczy napisać s.t.

$$\sum_{i \in M: b_i > 0} \sum_{j \in N: i > j} a_{ij} y_{ij} \geq d \quad \text{--- Suma warunkowa}$$

```
11 SumaWarunkowa:
12     sum {i in M, j in N:
13         b[i] > 0 and i > j} # Warunek
14         a[i, j] * y[i, j] >= d;
```

Nie trzeba też powtarzać **subject to** przy każdym ograniczeniu, wystarczy raz, przed pierwszym ograniczeniem.

$$\sum_{j \in N: a_{ij} > 0} a_{ij} y_{ij} \geq b_i, \quad \forall i \in M: b_i > 0 \quad \text{--- Warunkowa grupa ograniczeń}$$

```
15 OgraniczenieWarunkowe_SumaWarunkowa
16     {i in M:
17         b[i] > 0}: # Warunek
18         sum {j in N:
19             a[i, j] > 0} # Warunek
20             a[i, j] * y[i, j] >= b[i];
```